

```

*****
*   MACRO CODE – Risk-Based Monitoring Discussion           ;
*   NESUG 2013 9/9-9/11                                   ;
*   BOB HALL MS                                           ;
*****

%let nesug=27MAY2010;
libname datain "B:\CSP551\Backend Data\TestDump\CSP551_&nesug._Data"; * NESUG test;
libname tabout "B:\CSP551\Backend Data\TestDump\CSP551_&nesug._Data"; ** Output directory for tables';
proc format; value dcfagef 0='missing' 1='0-15' 2='16-30' 3='30-45' 4='45-60' 5='>60';
    value dcfstaf 1='Answered' 2='Closed' 0='Open';
    value colf 1='Site 1' 2='Site 2' 3='Site 3' 4='Site 4' 5='Site 5' 6='Site 6';
    value byf 1='VA' 2='Non-VA';
run;

*****
*   DCF Aging Data:                                       ;
*   - DCF Aging Element is aging_s_x                     ;
*****
data test; set datain.dl_dcfs_ns_&nesug.;
keep dcfstatus aging_s_x dcfagecat SiteGrpX SiteKeyx dcfstatusname2;
length dcfstatusname2 $ 15;
if aging_s_x eq . then dcfagecat = 0;
else if .lt aging_s_x le 15 then dcfagecat = 1;
else if 15 lt aging_s_x le 30 then dcfagecat = 2;
else if 30 lt aging_s_x le 45 then dcfagecat = 3;
else if 45 lt aging_s_x le 60 then dcfagecat = 4;
else if aging_s_x gt 60 then dcfagecat = 5;
if dcfstatusname2 = 'Ans' then dcfstatus = 1;
else if dcfstatusname2 = 'Clo' then dcfstatus = 2;
else dcfstatus = 0;
if SiteKeyx gt 3 then SiteKeyx = SiteKeyx - 3;
label SiteKeyx='Site Number';
label SiteGrpx='Site Type';
format dcfagecat dcfagef. dcfstatus dcfstaf.;
run;

*****
*   Format Table - Table of Report Elements               ;
*****
data tmp1;
input varname $14. order @18 varlabel $12. @34 type $2. +2 fformat $8.;
if fformat eq ' ' then fformat = '4.';
if _n_ gt 3 then varcat = 'DCF Status'; else varcat = 'DCF Age';
call symput('maxcount',_n_);
datalines;
aging_s_x    1 DCF Age Mean    n1
aging_s_x    1 DCF Age Median n2
dcfagecat    2 DCF Age      c2 dcfagef.
dcfstatus    3 DCF Status   c2 dcfstaf.
;
run;

```

```
*****,  
* Run tab1mac_docx Macro ;  
*****,
```

```
filename mprint 'B:\CSP551\Backend Data\TestDump\mprint.txt';  
options mprint symbolgen mlogic mfile;
```

```
%tab1mac_docx(tmp1,test, SitekeyX,SiteGrpX,tabout,Table1a);  
run;
```

```
*****,  
* ODS Listing for Final Report ;  
*****,
```

```
ods listing;  
ods tagsets.ExcelXP style=listing  
options (Sheet_Interval='proc' embedded_titles='Yes' Index='Yes'  
Absolute_Column_Width='20,20,20'  
Row_Heights='15,15,15,15,15,15,15')  
file="B:\CSP551\Backend Data\TestDump\Tables.xls";  
ods tagsets.ExcelXP options(sheet_name="Table 1a");  
proc document name=tabout.Table1a;  
replay ;  
run;  
ods tagsets.ExcelXP close;  
run;
```

```

*****
* Macro Coding (%tab1mac_docx )
*****

%macro tab1mac_docx(rdsn,dsn,colvar,byvar,libout,fileout);
ods listing close;
data results_f;
  dummy = 1;
  sorder = 0;
  &byvar = .;
  &colvar = .;
run;
data results_m;
  dummy = 1;
  sorder = 0;
  &byvar = .;
  &colvar = .;

run;
data results_m2;
  dummy = 1;
  sorder = 0;
  &byvar = .;
  &colvar = .;

run;
proc sort data=&dsn; by &byvar &colvar;
run;
%do count = 1 %to &maxcount;
  data tmpk1; set &rdsn;
  if _n_ eq &count;
  call symput('vcat',varcat);
  call symput('rf',varname);
  call symput('vartype',type);
  call symput('varlab',varlabel);
  call symput('vorder',order);
  call symput('fformat',fformat);
run;

  *** for categorical variables ***;
%if (&vartype eq c1) or (&vartype eq c2) %then %do;
  ods output crosstabfreqs=ctf ;
  proc freq data=&dsn;
%if &byvar eq %then %do;
  tables &rf * &colvar;
%end;
%else %do;
  tables &byvar * &rf * &colvar;
%end;
run;
quit;
data ctf; set ctf;
  length exposure $42;
  rename &rf=exposed;
  dichotomous = "&vartype";
  exposurecat = "&vcat";

```

```

if dichotomous eq 'c2' then exposure = catx(' ', "&varlab", put(&rf, &fformat));
else exposure = "&varlab";

if (dichotomous eq 'c1') then do;
  if (&colvar ne .) and (&rf eq 1) and ((_type_ eq '11') or (_type_ eq '111'));
end;
else do;
  if (&colvar ne .) and ((_type_ eq '11') or (_type_ eq '111'));
end;
keep &byvar &colvar exposurecat exposure &rf frequency colpercent sorder ;
sorder = &vorder;
run;
data results_f; set results_f ctf;
run;
%end;

```

```

*** for Continuous variables (mean and sd)***;

```

```

%if &vartype eq n1 %then %do;
proc means data=&dsn noprint; by &byvar &colvar;
  var &rf;
  output out= contout1 mean=mean std=std;
run;
data contout1; set contout1;
  length exposure $42;
  sorder = &vorder;
  exposurecat = "&vcats";
  exposure = "&varlab";
run;
data results_m; set results_m contout1;
run;
%end;

```

```

*** for Continuous variables (median and IQR)***;

```

```

%if &vartype eq n2 %then %do;
proc means data=&dsn noprint; by &byvar &colvar;
  var &rf;
  output out= contout2 median=median q1=q1 q3=q3;
run;
data contout2; set contout2;
  length exposure $42;
  sorder = &vorder;
  exposurecat = "&vcats";
  exposure = "&varlab";
  type="&vartype";
run;
data results_m2; set results_m2 contout2;
run;
%end;
%end;

```

```

*** clean up Frequencies ***;

```

```

data results_f; set results_f;
length freqpercent $30;
if dummy eq .;
colpercent = round(colpercent, .1);
if colpercent eq . then colpercent = 0;

```

```

freqpercent= cat(frequency,' (',colpercent,')');
run;

*** Clean up Means ***;
data results_m; set results_m;
  if dummy eq .;
length freqpercent $30;
  mean = round(mean,.01);
  std = round(std,.01);

freqpercent= cat(mean, ' (', std, ' )');
*** Clean up Medians ***;
data results_m2; set results_m2;
  if dummy eq .;
length freqpercent $30;
  median = round(median,.01);
  q1 = round(q1,.01);
  q3 = round(q3,.01);
freqpercent= cat(median, ' (', q1, ',', q3, ' )');
run;

%if &vcat ne %then %do;
proc sort data=results_f; by &byvar &colvar exposurecat sorder;
proc sort data=results_m; by &byvar &colvar exposurecat sorder;
proc sort data=results_m2; by &byvar &colvar exposurecat sorder;
data tabres; set results_f results_m results_m2; by &byvar &colvar exposurecat sorder;
keep exposurecat exposure &byvar &colvar freqpercent sorder frequency colpercent;
%end;
%else %do;
proc sort data=results_f; by &colvar sorder;
proc sort data=results_m; by &colvar sorder;
proc sort data=results_m2; by &colvar sorder;
data tabres; set results_f results_m results_m2; by &colvar sorder;
keep exposure &byvar &colvar freqpercent sorder frequency colpercent;
%end;

/*Label needs to be changed*/
label freqpercent='N(%)';
run;
proc print data=tabres;
run;

%if (&byvar ne ) and (&vcat ne ) %then %do;
ods listing;
ods document name=&libout.&fileout (write);
proc report data=tabres nowd headline ps=130;
column exposurecat exposure &byvar, &colvar, freqpercent ;
define exposurecat/ group order=data width=55;
define exposure /group order=data width=45;
define &byvar /across order=internal format=byf.;
define &colvar /across order=internal format=colf.;
define freqpercent /group width=20;
run;
ods document close;
run;

```

```

quit;
%end;

%else %if (&byvar ne ) and (&vcat eq ) %then %do;
ods listing;
ods document name=&libout.&fileout (write);
proc report data=tabres nowd headline ps=130 ;
column exposure &byvar, &colvar, freqpercent ;
define exposure /group order=data width=24;
define &byvar /across order=internal format=byf.;
define &colvar /across order=internal format=colf.;
define freqpercent /group width=20;
run;
ods document close;
run;
quit;
%end;

%else %if (&byvar eq ) and (&vcat ne ) %then %do;
ods listing;
ods document name=&libout.&fileout (write);
proc report data=tabres nowd headline ps=130 ;
column exposurecat exposure &colvar, freqpercent ;
define exposurecat/ group order=data width=55;
define exposure /group order=data width=45;
define &colvar /across order=internal format=colf.;
define freqpercent /group width=20;
run;
ods document close;
run;
quit;
%end;

%else %do;
ods listing;
ods document name=&libout.&fileout (write);
proc report data=tabres nowd headline ps=130 ;
column exposure &colvar, freqpercent ;
define exposure /group order=data width=45;
define &colvar /across order=internal format=colf.;
define freqpercent /group width=20;
run;
ods document close;
run;
quit;
%end;
run;
ods document close;
run;
quit;

proc freq data=&dsn;
tables &colvar;
run;
%mend tab1mac_docx;

```